********************************************************************************
*Project: Combating vaccine hesitancy: The case of HPV vaccination			   *
*Dataset used: HPV_Cali_database.dta		   								   *
*Author: Karina Marquez														   *
*Last update: May 2025													   	   *
*Software: Stata 17															   *
********************************************************************************

clear all
set more off
global workdir "C:\Users\karinama\OneDrive - Inter-American Development Bank Group\VPH Cali\VPH Cali - WP\Replication package - SSM"
global dbases "${workdir}\Data"
global tables "${workdir}\Tables"
global graphs "${workdir}\Graphs"
global output "${workdir}\Results"

********************************************************************************
*							 	RANDOMIZATION 								   *
********************************************************************************

use 	"${dbases}/HPV_Cali_database.dta", clear 	

*Stratification variables
  gen edad17 = (EdadID==17)
  gen ingreso=.
  replace ingreso=1 if NivelIngresoDescID=="nivel 1 del sisben" | NivelIngresoDescID=="nivel 2 del sisben" 
  replace ingreso=2 if NivelIngresoDescID=="Cotizantes con ingresos inferiores a 2 SMLMV" 
  replace ingreso=3 if NivelIngresoDescID=="Cotizantes con ingresos entre 2 y 5 SMLMV" | ///
  NivelIngresoDescID==	"Cotizantes con ingresos superiores a 5 SMLMV"

*Stratification
  egen strata=group(edad17 ingreso)

*Randomization
**Groups (6): control, placebo, T1 = information, T2 = social_norms, T3 = trust, T4 = framing
**You need to install: ssc install randtreat
  randtreat, gen(tratamiento) setseed(272022) strata(strata) multiple(6) misfits(global) replace

  la def tratamiento 0 "Control" 1 "Placebo" 2 "Information" 3 "Social norms" 4 "Trust" 5 "Framing" 
  la val tratamiento tratamiento

*Identifying subsidized individuals
  gen subsidiado=.
  replace subsidiado=1 if RolAfiliadoDescID=="Afiliado Subsidiado"
  replace subsidiado=0 if RolAfiliadoDescID=="Benificario"
  replace subsidiado=0 if RolAfiliadoDescID=="Cotizante"
  replace subsidiado=0 if RolAfiliadoDescID=="Upc Adicional"

  
********************************************************************************
*						VARIABLE CONSTRUCTION AND CLEANING 					   *
********************************************************************************

  tab tratamiento strata			//strata includes age and income
  tab tratamiento subsidiado		//subsidized individuals are those in levels 1 and 2 of the Sisben classification

*Identify who received each dose according to SURA records  
  gen dosis1_info_sura=1     if FechaVphDosis1ID!=.
  gen dosis2_info_sura=1     if FechaVphDosis2ID!=.
  gen dosis3_info_sura=1     if FechaVphDosis3ID!=.

*Dummy to identify girls vaccinated (any dose or date)
  gen	 d_vacunada=0				
  replace d_vacunada=1 if (FechaVphDosis1ID!=. | FechaVphDosis2ID!=. | FechaVphDosis3ID!=.) 
  *1,525 girls vaccinated [1,479 vaccinated through SURA]

  tab tratamiento if d_vacunada==1

*Income
  gen ingreso_bajo	= (ingreso==1)
  gen ingreso_medio	= (ingreso==2)
  gen ingreso_alto	= (ingreso==3)

*IPS
  encode  SucursalIpsDescID, gen(ips)	//Convert IPS from string to integer
  
*Label for stratified randomization groups
  label define GrupoStrata 	1 "Menor 17, Ingreso Bajo" 		///
							2 "Menor 17, Ingreso Medio" 	///
							3 "Menor 17, Ingreso Alto" 		///
							4 "Edad 17, Ingreso Bajo"   	///
							5 "Edad 17, Ingreso Medio"	 	///
							6 "Edad 17, Ingreso Alto"
  label values strata GrupoStrata
  tab 	strata 


*Timing of vaccine receipt
  **Treatment assignment was based on the December 9, 2022 dataset
  **Identify girls who received the vaccine before that date
  forval i=1(1)3 {
  	gen year_dosis`i' = year(FechaVphDosis`i'ID)
	gen month_dosis`i'= month(FechaVphDosis`i'ID)
	gen day_dosis`i'  = day(FechaVphDosis`i'ID)
  }
  
  
  **The intervention took place from January 10 to February 28, 2023
  **Identify those who received the vaccine before the intervention
  forval i=1(1)3 {
  	gen	    dosis`i'_prev_int = (year_dosis`i'<2023)
	replace dosis`i'_prev_int = 1 if year_dosis`i'==2023 & month_dosis`i'==1 & day_dosis`i'<10  	
  }
 
  gen  dosis_prev_int=(dosis1_prev_int==1 | dosis2_prev_int==1 | dosis3_prev_int==1)
  egen num_dosisprevint = rowtotal(dosis1_prev_int dosis2_prev_int dosis3_prev_int)

  
  **Identify if some had completed the full vaccination schedule before the intervention
  **A complete schedule, as defined by SURA in Colombia, consists of at least 2 doses
  gen 	  todasvacuna_prev_int = (dosis1_prev_int==1 & dosis2_prev_int==1 ///
								  & dosis3_prev_int==1)
  replace todasvacuna_prev_int = 1 if num_dosisprevint==2 & todasvacuna_prev_int==0
							  
  tab todasvacuna_prev_int

  
  **Identify those who received the vaccine beginning on January 10, 2023, the start of the intervention
  **Information is only available through 2023
  gen 	  vacunada_int_d1 = 1 if year_dosis1==2023 & month_dosis1>1 & FechaVphDosis1ID!=.	
  replace vacunada_int_d1 = 1 if year_dosis1==2023 & month_dosis1==1 & day_dosis1>=10 & 	///
								 FechaVphDosis1ID!=. & vacunada_int_d1==.
  replace vacunada_int_d1 = 0 if vacunada_int_d1==.
	
  gen 	  vacunada_int_d2 = 1 if year_dosis2==2023 & month_dosis2>1 & FechaVphDosis2ID!=.	
  replace vacunada_int_d2 = 1 if year_dosis2==2023 & month_dosis2==1 & day_dosis2>=10 & 	///
								 FechaVphDosis2ID!=. & vacunada_int_d2==.
  replace vacunada_int_d2 = 0 if vacunada_int_d2==.
	
  gen 	  vacunada_int_d3 = 1 if year_dosis3==2023 & month_dosis3>1 & FechaVphDosis3ID!=.	
  replace vacunada_int_d3 = 1 if year_dosis3==2023 & month_dosis3==1 & day_dosis3>=10 & 	///
								 FechaVphDosis3ID!=. & vacunada_int_d3==.
  replace vacunada_int_d3 = 0 if vacunada_int_d3==. 
	 
  gen  	  vacunada_int = (vacunada_int_d1==1 | vacunada_int_d2==1 | vacunada_int_d3==1)
  replace vacunada_int = . if todasvacuna_prev_int==1
  
	
********************************************************************************
*									BALANCE 								   *
********************************************************************************

*Descriptive statistics
  label var subsidiado			"Subsidized"
  label var EdadID 				"Age"
  label var edad17 				"17 years old"
  label var ingreso_bajo 		"Low income"
  label var ingreso_medio 		"Medium income"
  label var ingreso_alto 		"High income"
  label var dosis_prev_int 		"At least one dose"
  label var todasvacuna_prev_int "Complete scheme"

  global balance subsidiado EdadID edad17 ingreso_bajo ingreso_medio ingreso_alto  ///
  dosis_prev_int todasvacuna_prev_int
 
  eststo total_means:   estpost summarize $balance 
  eststo control_means: estpost summarize $balance if tratamiento == 0
  eststo placebo_means: estpost summarize $balance if tratamiento == 1
  eststo info_means:    estpost summarize $balance if tratamiento == 2
  eststo norms_means:   estpost summarize $balance if tratamiento == 3
  eststo trust_means:   estpost summarize $balance if tratamiento == 4
  eststo framing_means: estpost summarize $balance if tratamiento == 5
 
*Balance
  esttab total_means control_means placebo_means info_means norms_means trust_means  ///
  framing_means using "${tables}/T1 descriptive statistics.tex", replace 			 ///
  cell(mean(pattern(1 1 1 1 1 1 1) fmt(3)) 										     ///
  sd(pattern(0 1 1 1 1 1 1) fmt(3) par)) 											 ///
  nomtitles 																		 /// 
  prehead(`"\begin{table}[htbp]"'	`"\centering"' `"\footnotesize"'				 ///
			`"\caption{\centering Descriptive statistics} \label{bal}"'			  	 ///
			`"\begin{tabular}{lccccccc}"' `"\toprule"'								 ///	
			`"&\multicolumn{1}{c}{Total}&\multicolumn{6}{c}{Means (s.d.)} \\"' 		 ///  
			`"\cmidrule{3-8}"' 						   								 ///  
			`"&\multicolumn{1}{c}{sample} & \multicolumn{1}{c}{Control} &"' 		 /// 
			`"\multicolumn{1}{c}{Placebo} & \multicolumn{1}{c}{Information} &"' 	 ///
			`"\multicolumn{1}{c}{Social norms} & \multicolumn{1}{c}{Trust} &"'		 ///
			`"\multicolumn{1}{c}{Framing} \\"' 										 ///  
			`"& & (C) & (P) & (T\textsubscript{1}) & (T\textsubscript{2})"'   		 ///
			`"& (T\textsubscript{3}) & (T\textsubscript{4}) \\"') 				   	 ///
  label collabels(none) 															 ///  
  refcat(ingreso_bajo " \textbf{Income} \vspace{.3em}" 							     ///
		 dosis_prev_int "   \\  \textbf{Previous doses} \vspace{.3em}", label(" "))  ///
  stats(N, fmt(%15.0fc) labels("Total observations")) nonum 						 ///
  postfoot(`"\hline"' `"\addlinespace"' `"\multicolumn{8}{p{\linewidth}}{\footnotesize Notes: Means and standard deviations (SD) are presented. SD are in parenthesis. Previous doses indicates if the girl has received any dose before the intervention began.} \\"' `"\end{tabular}"' `"\end{table}"')
	 
	 
*Comparison across treatment groups
  forval i=1(1)5 {
	eststo diff1_`i':  estpost ttest $balance if tratamiento==`i' | tratamiento==0, by(tratamiento) unequal 
  }
  forval i=2(1)5 {
	eststo diff2_`i':  estpost ttest $balance if tratamiento==`i' | tratamiento==1, by(tratamiento) unequal 
  }
  forval i=3(1)5 {
	eststo diff3_`i':  estpost ttest $balance if tratamiento==`i' | tratamiento==2, by(tratamiento) unequal 
  }
  forval i=4(1)5 {
	eststo diff4_`i':  estpost ttest $balance if tratamiento==`i' | tratamiento==3, by(tratamiento) unequal 
  }
	eststo diff5_4:    estpost ttest $balance if tratamiento==5 | tratamiento==4, by(tratamiento) unequal 

  **Table
	esttab diff1_* diff2_* diff3_* diff4_* diff5_4 									 	   ///
	using "${tables}/TA1 balance comparison.tex", replace 							 	   ///
	cell(b(pattern(1 1 1 1 1 1 1 1 1 1 1 1 1 1 1) fmt(3) star) 							   ///
	se(pattern(1 1 1 1 1 1 1 1 1 1 1 1 1 1 1) fmt(3) par)) 							 	   ///
	mtitle(	"C - P" "C - T\$_{1}\$" "C - T\$_{2}\$" "C - T\$_{3}\$" "C - T\$_{4}\$"  	   ///
			"P - T\$_{1}\$" "P - T\$_{2}\$" "P - T\$_{3}\$" "P - T\$_{4}\$" 		 	   ///
			"T\$_{1}\$ - T\$_{2}\$" "T\$_{1}\$ - T\$_{3}\$" "T\$_{1}\$ - T\$_{4}\$"  	   ///
			"T\$_{2}\$ - T\$_{3}\$" "T\$_{2}\$ - T\$_{4}\$" "T\$_{3}\$ - T\$_{4}\$") 	   ///
	label collabels(none) nonum 													 	   ///
	refcat(	ingreso_bajo " \textbf{Income} \vspace{.3em}" 							 	   ///
			dosis_prev_int "  \\ \textbf{Previous doses} \vspace{.3em}", label(" ")) 	   ///
	prehead(`"\begin{table}[htbp]"'	`"\tiny"' `"\centering"' 		   				 	   ///
			`"\caption{\centering Balance - Comparison among treatments} \label{balcomp}"' ///
			`"\begin{tabular}{lccccccccccccccc}"' `"\toprule"')						 	   ///
	postfoot(`"\hline"' `"\addlinespace"' `"\multicolumn{16}{p{80.9em}}{\tiny Notes: Previous doses indicates whether the girl has received any dose before the intervention. A complete scheme is conformed by two doses. C is the control group, P is the placebo group, \$T_{1}\$ is the Information SMS group, \$T_{2}\$ is the Social Norms SMS group, \$T_{3}\$ is the Trust SMS groupand \$T_{4}\$ is the Framing SMS group. Standard errors in parentheses. *** p$<$0.001, ** p$<$0.01, * p$<$0.05} \\"' `"\end{tabular}"' `"\end{table}"') substitute("\_" "_") ///
	starlevels(* .05 ** .01 *** .001) 
	 
 
*Vaccination rate by treatment
  preserve
	gen vac_int=vacunada_int if vacunada_int==1
	egen pc_vac = mean(100*(vacunada_int==1)), by (tratamiento)

	label var tratamiento	"Total number of girls"
	label var vac_int		"Total number of vaccinated girls"
	label var pc_vac		"Vaccination rate (\%)"

	global response_rate tratamiento vac_int pc_vac 

	eststo control_rr: estpost sum $response_rate if tratamiento == 0
	eststo placebo_rr: estpost sum $response_rate if tratamiento == 1
	eststo info_rr:    estpost sum $response_rate if tratamiento == 2
	eststo norms_rr:   estpost sum $response_rate if tratamiento == 3  
	eststo trust_rr:   estpost sum $response_rate if tratamiento == 4
	eststo framing_rr: estpost sum $response_rate if tratamiento == 5

	  **Table 
		esttab control_rr placebo_rr info_rr norms_rr trust_rr framing_rr  			///
		using "${tables}/T rates.tex", replace 			   				///
		cell(count(pattern(1 1 1 1 1 1) fmt(0) keep(tratamiento vac_int)) 			///
			 mean(pattern(1 1 1 1 1 1) fmt(2) keep(pc_vac))) 						///			
		nomtitles 																	/// 
		prehead(`"\begin{table}[htbp]"'	`"\centering"' `"\footnotesize"'			///
				`"\caption{\centering Vaccination Rate by Treatment} \label{rates}"'		///
				`"\begin{tabular}{p{14em}cccccc}"' `"\toprule"'						///	
				`"& \multicolumn{1}{c}{Control} & "' 		 						/// 
				`"\multicolumn{1}{c}{Placebo} & \multicolumn{1}{c}{Information} &"' ///
				`"\multicolumn{1}{c}{Social norms} & \multicolumn{1}{c}{Trust} &"'	///
				`"\multicolumn{1}{c}{Framing} \\"' 									///  
				`"& (C) & (P) & (T\textsubscript{1}) & (T\textsubscript{2})"'   	///
				`"& (T\textsubscript{3}) & (T\textsubscript{4}) \\"') 				///
		label collabels(none)	stats(, fmt(%15.0fc)) nonum 						///
		refcat(	pc_vac " \hline ", nolabel) 			///
		postfoot(`"\addlinespace"' `"\multicolumn{7}{p{\linewidth}}{} \\"' `"\end{tabular}"' `"\end{table}"')
  restore

********************************************************************************
*								  REGRESSIONS 								   *
********************************************************************************

*P-value stars
  capture program drop stars  
  program define stars  
	args pvalue  
	local stars = cond(`pvalue'<0.001, "***", cond(`pvalue'<0.01, "**", cond(`pvalue'<0.05, "*", "")))  
	c_local stars "`stars'"  
  end


*Main model
  reg vacunada_int i.tratamiento
  estimates store Model1
  outreg2 using "${output}/reg_modelobase.xls", replace ctitle(Model 1) label dec(3) alpha(0.001, 0.01, 0.05)

  reg vacunada_int i.tratamiento dosis_prev_int subsidiado i.ips  i.strata 
  estimates store Model2
  outreg2 using "${output}/reg_modelobase.xls", append ctitle(Model 2) label dec(3) alpha(0.001, 0.01, 0.05)
  erase "${output}\reg_modelobase.txt"

  **Graph
	coefplot (Model2), keep(*.tratamiento) yline(0) vertical title("") 							///
	ytitle("Percent")  graphregion(color(white)) rescale(100) 									///
	msymbol(d) mfcolor(white) levels(99.9 99 95) xsize(28) ysize(20)							///
	ciopts(color(teal lavender navy))															///
	note("Notes: The figure shows the marginal effect of each treatment compared to the" 	  	///
		 "control group. The control group effect is 5.8 p.p., represented by the horizontal" 	///
		 "line. It includes 99.9%, 99% and 95% confidence intervals.", span margin(3 3 3 3)   	///
		 size(medium)) 
	graph export "${graphs}/F2 Coeficientes por tratamiento.png", replace width(2400) height(1600)

  **Tabla 2
	texdoc do "${workdir}\Do files\Table 2 - main model.do"

	
*Heterogeneities

  **Heterogeneity by number of dose received
	reg vacunada_int i.tratamiento##i.dosis_prev_int 
	margins tratamiento#dosis_prev_int, post
	outreg2 using "${output}/reg_het_dosis_margins.xls", replace ctitle(Model 1) ///
	label dec(3) stats(coef se pval ci) alpha(0.001, 0.01, 0.05)
	
	reg vacunada_int i.tratamiento##i.dosis_prev_int subsidiado i.ips  i.strata
	margins tratamiento#dosis_prev_int, post
	marginsplot, graphregion(color(white))  title("") ytitle("")					///
	legend(order(1 "First dose" 2 "Second dose") size(small)) plotopts(connect(i))	///
	ylabel(, labsize(small)) horizontal xlabel(, labsize(small)) 					///
	xtitle(,size(small))  xline(0, lwidth(vthin) lpattern(dash) lcolor(black))		///
	ciopt(lpattern(shortdash)) 														///
	note("Notes: The figure shows the predicted mean vaccination rates by treatment and whether the girl had received" 	 	  ///
		 "a dose before the intervention. It includes 95% confidence interval. Estimates are based on a linear regression"    ///
		 "model with an interaction term for treatment and prior vaccination status, including covariates such as subsidized" ///
		 "scheme status, healthcare center visited, and the stratification variable which includes level of income and age.", ///
		 span size(2.6) margin(3 3 3 3))
	graph export "${graphs}/F3 het dosis.png", replace
	outreg2 using "${output}/reg_het_dosis_margins.xls", drop(i.ips) append 	///
	ctitle(Model 2) label dec(3) stats(coef se pval ci) alpha(0.001, 0.01, 0.05)
	erase "${output}/reg_het_dosis_margins.txt"


	**Table
	texdoc do "${workdir}\Do files\Table A2 - het dose.do"
	
	
  **Heterogeneity by age (17 or older)
	reg vacunada_int i.tratamiento##i.edad17
	margins tratamiento#edad17, post
    outreg2 using "${output}/reg_het_edad_margins.xls", replace ctitle(Model 1) ///
	label dec(3) stats(coef se pval ci) alpha(0.001, 0.01, 0.05)
  
	reg vacunada_int i.tratamiento##i.edad17 dosis_prev_int subsidiado i.ips ingreso
	margins tratamiento#edad17, post
	marginsplot, graphregion(color(white)) title("") ytitle("") horizontal 				///
	xtitle(, size(small)) plotopts(connect(i)) ylabel(, labsize(small)) 				///
	legend(order(1 "Under 17 y.o." 2 "17 y.o.") size(small)) xlabel(,labsize(small)) 	///
	ciopt(lpattern(shortdash)) xline(0, lwidth(vthin) lpattern(dash) lcolor(black)) 	///
	note("Notes: The figure shows the predicted mean vaccination rates by treatment type and whether the girl"     ///
		 "is under or exactly 17 years old, with 95% confidence intervals. The estimates are based on a linear"    ///
		 "regression model that includes an interaction term for treatment and age group,along with covariates"    ///
		 "such as subsidized scheme status, whether a girl has received a dose before, healthcare center visited," ///
		 "and household income level.", span size(2.75) margin(3 3 3 3))
	graph export "${graphs}/FA1 het age.png",  replace
    outreg2 using "${output}/reg_het_edad_margins.xls", drop(i.ips) append 		///
	ctitle(Model 2) label dec(3) stats(coef se pval ci) alpha(0.001, 0.01, 0.05)
	erase "${output}/reg_het_edad_margins.txt"
 	
	
	**Table
	texdoc do "${workdir}\Do files\Table A3 - het age.do"


  **Heterogeneity by income level
 	reg vacunada_int i.tratamiento##i.ingreso
	margins tratamiento#ingreso, post
	outreg2 using "${output}/reg_het_inc_margins.xls", replace ctitle(Model 1)  ///
	label dec(3) stats(coef se pval ci)  alpha(0.001, 0.01, 0.05)
	
	reg vacunada_int i.tratamiento##i.ingreso i.ips dosis_prev_int edad17
	margins tratamiento#ingreso, post 
	marginsplot, graphregion(color(white)) title("") ytitle("") horizontal 				///
	plotopts(connect(i)) xtitle(,size(small)) ciopt(lpattern(shortdash))				///
	legend(order(1 "Low income" 2 "Medium income" 3 "High income") cols(3) size(small)) ///
	ylabel(, labsize(small)) xlabel(, labsize(small))  									///
	xline(0, lwidth(vthin) lpattern(dash) lcolor(black))									///
	note("Notes: The figure shows the predicted mean vaccination rates by treatment and household income level"    ///
		 "It includes 95% confidence interval. Estimates are based on a linear regression model incorporating an"  ///
		 "interaction term for treatment and household income level, along with covariates such as whether a girl" ///
		 "has received a dose before, healthcare center visited and girl's age.", span size(2.75) margin(3 3 3 3))
	graph export "${graphs}/FA2 het income.png",   replace
	outreg2 using "${output}/reg_het_inc_margins.xls", drop(i.ips) append 		///
	ctitle(Model 2) label dec(3) stats(coef se pval ci) alpha(0.001, 0.01, 0.05)
	erase "${output}/reg_het_inc_margins.txt" 
	
	**Table
	texdoc do "${workdir}\Do files\Table A4 - het income.do"
	
	 
********************************************************************************
*							 MESSAGES CONTENT  							   	   *
********************************************************************************
	
*Spanish
  **Table A5
  	file open myfile using "${tables}/TA5 messages SPA.tex", write replace  
	{ 
	file write myfile "\begin{landscape}" _n  
	file write myfile "\begin{table}[htbp]" _n  
	file write myfile "  \centering" _n  
	file write myfile "  \tiny" _n  
	file write myfile "  \caption{\centering Content of messages (Spanish)} \label{messages_spa}" _n  
	file write myfile "   \begin{tabular}{lp{8.285em}p{8.285em}p{8.285em}p{8.285em}p{8.285em}p{10em}p{8.285em}p{8.285em}}" _n  
	file write myfile "   \hline" _n  
	file write myfile "   \multicolumn{1}{p{6.145em}}{\textbf{Treatment}} & \textbf{Message 1} & \textbf{Message 2} & \textbf{Message 3} & \textbf{Message 4} & \textbf{Message 5} & \textbf{Message 6} & \textbf{Message 7} & \textbf{ Message 8} \\" _n  
	file write myfile "   \hline" _n
	
	*Control
	file write myfile "   \multicolumn{1}{p{6.145em}}{Control} & No message & No message & No message & No message & No message & No message & No message & No message \\" _n  
	file write myfile "   \hline" _n  
	
	*Placebo
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Placebo}} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, te brindamos todo lo que tu familia necesita.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, juntos por un Cali saludable.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, la salud es lo más importante en la vida. \newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, cuida tu salud.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, contamos con los servicios de salud que tu familia necesita.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, felices fiestas. Cuida tu salud.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, feliz año nuevo. Sigue los lineamientos de salud este nuevo año.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, feliz año nuevo. Mantente saludable en este nuevo año.\newline{} \\" _n  
	file write myfile "       & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud \\" _n  
	file write myfile "   \hline" _n  

	*Information
	file write myfile "   \multirow{2}[2]{*}{Information} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, sabías que la vacuna contra el VPH es gratuita en Cali para las niñas de 9-17? & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que las niñas de 9-17 años deben vacunarse contra el VPH.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, hay 150 centros de vacunación públicos en Cali en donde puedes vacunar a tu hija contra el VPH. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, la vacunación contra el VPH reduce 89\% el riesgo de cáncer cervical.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, vacuna a tu hija contra el VPH. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, haz click aquí y ubica los sitios de vacunación contra el VPH en Cali: https://bit.ly/ ssaludcali. Vacuna a tu hija.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, tu hija de 9-17 años tiene derecho s vacunarse gratuitamente contra el VPH. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que a tu hija aún le falta la vacuna contra el VPH.\newline{} \\" _n  
	file write myfile "       & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud \\" _n  
	file write myfile "   \hline" _n  
	  
	*Social norms
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Social Norms}} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, 3 de 10 padres en Cali con niñas de edad de tu hija ya vacunaron a su hija contra el VPH. Todavía faltas tú :( & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que las niñas de 9-17 deben vacunarse contra el VPH y la vacuna es gratuita para esas edades. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, en Cali 3 de 10 padres de niñas como tu hija ya las han vacunado contra el VPH y cada vez se suman más. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, la vacunación contra el VPH reduce 89\% el riesgo de cáncer cervical. Vacúnala en uno de los 150 centros de vacunación. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, 3 de 10 padres en Cali con niñas de edad de tu hija las han vacunado contra el VPH, un alza del 149\% desde 2016. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, haz click aquí y ubica los sitios de vacunación contra el VPH en Cali: https://bit.ly/ ssaludcali. Vacuna a tu hija.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_,3 de 10 padres de niñas de 9-17 años ya las han vacunado contra el VPH, un alza del 149\% desde 2016. Todavía faltas tú :( & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que a tu hija aún le falta la vacuna contra el VPH. La vacunación es gratis para las niñas de 9-17.\newline{} \\" _n  
	file write myfile "       & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud \\" _n  
	file write myfile "   \hline" _n  
	  
	*Trust
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Trust}} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, los médicos especialistas recomiendan la vacunación de tu hija contra el VPH. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que las niñas de 9-17 deben vacunarse contra el VPH y la vacuna es gratuita para esas edades. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, la Secretaría de Salud en Cali recomieda la vacunación de tu hija contra el VPH.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, la vacunación contra el VPH reduce 89\% el riesgo de cáncer cervical. Vacúnala en uno de los 150 centros de vacunación. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, Cali recomienda y pone a disposición la vacuna del VPH para el bienestar de tu hija en 150 centros de vacunación.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, haz click aquí y ubica los sitios de vacunación contra el VPH en Cali: https://bit.ly/ ssaludcali. Vacuna a tu hija.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, los médicos y la Secretaría de Salud firmemente recomiendan la vacunación contra el VPH de tu hija y niñas de 9-17 años.  & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que a tu hija aún le falta la vacuna contra el VPH. La vacunación es gratis para las niñas de 9-17.\newline{} \\" _n  
	file write myfile "       & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud \\" _n  
	file write myfile "   \hline" _n  
	  
	*Framing
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Framing}} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, tu hija debe tener puestas las 21 vacunas en su carnet de vacunación y aún le falta la vacuna contra el VPH.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que las niñas de 9-17 deben vacunarse contra el VPH y la vacuna es gratuita para esas edades. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, Sabes que la vacuna contra el VPH es tan segura como otras vacunas? & Hola mamá de \_\_\_\_\_\_\_\_\_\_, la vacunación contra el VPH reduce 89\% el riesgo de cáncer cervical. Vacúnala en uno de los 150 centros de vacunación. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, tienes una cita el 00 de MES a la HORA am para vacunar a tu hija contra el VPH. Hay una vacuna reservada para ella. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, haz click aquí y ubica los sitios de vacunación contra el VPH en Cali: https://bit.ly/ ssaludcali. Vacuna a tu hija.\newline{} & Hola mamá de \_\_\_\_\_\_\_\_\_\_, en el 2020, 4,742 mujeres desarrollaron cáncer cervical en Cali. Evita el cáncer con la vacuna de VPH en niñas de 9-17. & Hola mamá de \_\_\_\_\_\_\_\_\_\_, recuerda que a tu hija aún le falta la vacuna contra el VPH. La vacunación es gratis para las niñas de 9-17.\newline{} \\" _n  
	file write myfile "       & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud & Secretaría de Salud \\" _n  
	file write myfile "   \hline" _n  

	file write myfile "   \end{tabular}%" _n  
	file write myfile "  \label{tab:addlabel}%" _n  
	file write myfile "  \end{table}%" _n  
	file write myfile "\end{landscape}" _n  
	} 
	file close myfile
		

*Inglés: 
  **Tabla A6
	file open myfile using "${tables}/TA6 messages ENG.tex", write replace  
	{
	file write myfile "\begin{landscape}" _n  
	file write myfile "\begin{table}[htbp]" _n  
	file write myfile "  \centering" _n  
	file write myfile "  \tiny" _n  
	file write myfile "  \caption{\centering Content of messages (English)} \label{messages_eng}" _n  
	file write myfile "   \begin{tabular}{lp{8.285em}p{8.285em}p{8.285em}p{8.285em}p{8.285em}p{10em}p{8.285em}p{8.285em}}" _n  
	file write myfile "   \hline" _n  
	file write myfile "   \multicolumn{1}{p{6.145em}}{\textbf{Treatment}} & \textbf{Message 1} & \textbf{Message 2} & \textbf{Message 3} & \textbf{Message 4} & \textbf{Message 5} & \textbf{Message 6} & \textbf{Message 7} & \textbf{ Message 8} \\" _n  
	file write myfile "   \hline" _n   	

	* Control   
	file write myfile "   \multicolumn{1}{p{6.145em}}{Control} & No message & No message & No message & No message & No message & No message & No message & No message \\" _n  
	file write myfile "   \hline" _n  
	  
	* Placebo   
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Placebo}} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, we offer you everything your family needs.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, together for a healthy Cali.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, health is the most important thing in life. \newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, take care of your health.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, we have the health services your family needs.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, Happy Holidays. Take care of your health.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, Happy New Year. Follow the health guidelines this new year.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, Happy New Year. Stay healthy this new year.\newline{} \\" _n  
	file write myfile "       & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health \\" _n  
	file write myfile "   \hline" _n  
	  
	* Information   
	file write myfile "   \multirow{2}[2]{*}{Information} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, did you know that the HPV vaccine is free in Cali for girls aged 9-17? & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that girls aged 9-17 should get vaccinated against HPV.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, there are 150 public vaccination centers in Cali where you can vaccinate your daughter against HPV. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, the HPV vaccine reduces the risk of cervical cancer by 89\%.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, vaccinate your daughter against HPV. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, click here to find the HPV vaccination centers in Cali: https://bit.ly/ssaludcali. Vaccinate your daughter.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, your daughter aged 9-17 has the right to get vaccinated against HPV for free. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that your daughter still needs the HPV vaccine.\newline{} \\" _n  
file write myfile "     & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health \\" _n 
	file write myfile "   \hline" _n  
	  
	* Social Norms   
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Social Norms}} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, 3 out of 10 parents in Cali with daughters of your daughter's age have already vaccinated their daughters against HPV. You're still missing :( & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that girls aged 9-17 should get vaccinated against HPV and the vaccine is free for those ages. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, in Cali, 3 out of 10 parents of girls like your daughter have already vaccinated them against HPV and more are joining every day. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, the HPV vaccine reduces the risk of cervical cancer by 89\%. Vaccinate her at one of the 150 vaccination centers. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, 3 out of 10 parents in Cali with daughters of your daughter's age have vaccinated them against HPV, a 149\% increase since 2016. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, click here to find the HPV vaccination centers in Cali: https://bit.ly/ssaludcali. Vaccinate your daughter.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, 3 out of 10 parents of girls aged 9-17 have already vaccinated them against HPV, a 149\% increase since 2016. You're still missing :( & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that your daughter still needs the HPV vaccine. Vaccination is free for girls aged 9-17.\newline{} \\" _n  
file write myfile "     & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health \\" _n  
	file write myfile "   \hline" _n  
	  
	* Trust   
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Trust}} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, specialist doctors recommend vaccinating your daughter against HPV. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that girls aged 9-17 should get vaccinated against HPV and the vaccine is free for those ages. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, the Secretary of Health in Cali recommends vaccinating your daughter against HPV.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, the HPV vaccine reduces the risk of cervical cancer by 89\%. Vaccinate her at one of the 150 vaccination centers. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, Cali recommends and provides the HPV vaccine for the well-being of your daughter at 150 vaccination centers.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, click here to find the HPV vaccination centers in Cali: https://bit.ly/ssaludcali. Vaccinate your daughter.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, doctors and the Secretary of Health strongly recommend vaccinating your daughter and girls aged 9-17 against HPV. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that your daughter still needs the HPV vaccine. Vaccination is free for girls aged 9-17.\newline{} \\" _n  
file write myfile "     & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health \\" _n  
	file write myfile "   \hline" _n  
	  
	* Framing   
	file write myfile "   \multicolumn{1}{l}{\multirow{2}[2]{*}{Framing}} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, your daughter should have all 21 vaccines in her vaccination card and she's still missing the HPV vaccine.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that girls aged 9-17 should get vaccinated against HPV and the vaccine is free for those ages. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, did you know that the HPV vaccine is as safe as other vaccines? & Hello \_\_\_\_\_\_\_\_\_\_'s mom, the HPV vaccine reduces the risk of cervical cancer by 89\%. Vaccinate her at one of the 150 vaccination centers. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, you have an appointment on the 00th of MONTH at TIME am to vaccinate your daughter against HPV. There's a vaccine reserved for her. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, click here to find the HPV vaccination centers in Cali: https://bit.ly/ssaludcali. Vaccinate your daughter.\newline{} & Hello \_\_\_\_\_\_\_\_\_\_'s mom, in 2020, 4,742 women developed cervical cancer in Cali. Prevent cancer with the HPV vaccine for girls aged 9-17. & Hello \_\_\_\_\_\_\_\_\_\_'s mom, remember that your daughter still needs the HPV vaccine. Vaccination is free for girls aged 9-17.\newline{} \\" _n  
file write myfile "     & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health & Secretariat of Health \\" _n
	file write myfile "   \hline" _n  
	
	file write myfile "   \end{tabular}%" _n  
	file write myfile "  \label{tab:addlabel}%" _n  
	file write myfile "  " _n  
	file write myfile "\end{table}%" _n  
	file write myfile "\end{landscape}" _n  
	}
	file close myfile	